

# Технологическая карта для проекта «Фрактал Мандельброта»

Морозов В. В.

Множество Мандельброта — это множество таких точек  $z$  на комплексной плоскости, для которых итерационная последовательность  $z_{n+1}=z_n^2+c$  при  $z_0=0$  является ограниченной. То есть, это множество таких  $c$ , для которых существует такое действительное  $R$ , что неравенство  $|z_n|<R$  выполняется при всех натуральных  $n$ .

Множество Мандельброта является одним из самых известных фракталов, в том числе за пределами математики, благодаря своим цветным визуализациям. Его фрагменты не строго подобны исходному множеству, но при многократном увеличении определённые части всё больше похожи друг на друга<sup>1</sup>.

Предлагаемая технологическая карта поможет учащимся разработать в среде Delphi программу для построения фрактала Мандельброта. Программа позволит пользователю сколько угодно подробно рассматривать и масштабировать фрактал Мандельброта.

**1. Запустите Delphi, сохраните новый проект в папке Фрактал, сохраните файлы проекта как fract.pas и fractal.dpr.**

**2. Измените некоторые свойства формы:**

Name	frmMain
Caption	Фрактал Мандельброта
Icon	Поставьте свою иконку
WindowState	WsNormal
Width	355
Height	315
BorderStyle	bsDialog (чтобы нельзя было изменять размер окна)

**3. Разместите на форме следующие объекты и измените их свойства:**

**А) Объект Image.**

Name	Image
Width	250
Height	250

**Б) Бросьте на форму 5 объектов Edit;**

Name	Edit1, Edit2, Edit3, Edit4, Edit5
Text	-1,5; 0,5; -1; 1; Фрактал Первые 4 объекта для хранения чисел, 5-й объект – для хранения имени сохраняемого файла

**В) Бросьте на форму 4 метки для подписи объектов Edit. Подпишите их метками так:**

Объект	Подпись меткой	В эти ячейки мы будем вписывать пределы изменения аргумента по оси $x$  И по оси $y$ .
Edit1	Pmin	
Edit2	Pmax	
Edit3	Qmin	
Edit4	Qmax	

<sup>1</sup> Wikipedia.

### Г) Бросьте на форму 2 кнопки:

Name	btnDraw Кнопка будет запускать процесс рисования фрактала	Name	btnSave Кнопка будет сохранять готовый фрактал в папку с программой
Caption	Рисовать	Caption	Сохранить

### Д) Бросьте на форму объект **ProgressBar** и измените его имя как **ProgressBar**.

Поместите **ProgressBar** точно под **Image**. Этот объект будет показывать, что компьютер не завис, и процесс идёт, будет показывать, на сколько фрактал уже построен и сколько ещё ждать.

### 4. Напишем обработчик нажатия на кнопку **РИСОВАТЬ**. Не забудьте, что вам нужно набрать только переменные и код между бегинном и эндом. Не нужно писать заголовок процедуры вручную!!

```
procedure TfrmMain.btnDrawClick(Sender: TObject);
var i, j, t, time, size, L: integer;
    p0, q0, p1, q1, p, q, x, y, xx: double;
begin
    time:=500; size:=250; L:=100;
    p0:=StrToFloat(Edit1.Text); p1:=StrToFloat(Edit2.Text);
    q0:=StrToFloat(Edit3.Text); q1:=StrToFloat(Edit4.Text);
    for i:=0 to size-1 do
        begin
            p:=p0+i*(p1-p0)/(size-1);
            for j:=0 to size-1 do
                begin
                    q:=q0+j*(q1-q0)/(size-1);
                    x:=0; y:=0; t:=0;
                    repeat
                        inc(t);
                        xx:=x;
                        x:=x*x-y*y+p;
                        y:=2*xx*y+q;
                    until (x*x+y*y>L) or (t=time);
                    if t=time
                        then
                            //Image.Canvas.Pixels[i, j]:=rgb(j, i, i)
                            Image.Canvas.Pixels[i, j]:=clNavy
                        else
                            //Image.Canvas.Pixels[i, j]:=
                                //RGB((250+t*t*t div 2) div 256+i,
                                //(250+t*t div 2) div 256+j, t div 2)
                            Image.Canvas.Pixels[i, j]:=
                                RGB(250-t div 2, 250-t div 2, 250)
                end;
            ProgressBar.Position:=round(i*100/size)
        end;
    ProgressBar.Position:=0
end;
```

### 5. Напишем обработчик нажатия на кнопку СОХРАНИТЬ

```
procedure TfrmMain.btnSaveClick(Sender: TObject);
begin
    image.Picture.SaveToFile(edit5.text+'.bmp')
end;
```

### 6. Напишем процедуру, которая при клике на объекте Image приблизит картинку в 4 раза. Для этого у объекта Image в обработчике событий выберите событие OnMouseDown.

```
procedure TfrmMain.ImageMouseDown(Sender: TObject; Button:
TMouseButton;
    Shift: TShiftState; X, Y: Integer);
var p0, q0, p1, q1, x0, y0, x1, y1: double;
begin
    p0:=StrToFloat(Edit1.Text); p1:=StrToFloat(Edit2.Text);
    q0:=StrToFloat(Edit3.Text); q1:=StrToFloat(Edit4.Text);
    if button=mbLeft
    then
        begin
            x0:=p0+(p1-p0)*x/Image.Width-(p1-p0)/4;
            y0:=q0+(q1-q0)*y/Image.Height-(q1-q0)/4;
            x1:=p0+(p1-p0)*x/Image.Width+(p1-p0)/4;
            y1:=q0+(q1-q0)*y/Image.Height+(q1-q0)/4;
            image.Canvas.pen.Color:=clred;
            image.Canvas.brush.Style:=bsclear;
            image.Canvas.Rectangle(x-image.width div 4, Y-image.Height
div 4, x+image.width div 4, y+image.height div 4);
            image.Repaint;
        end
    else
        begin
            x0:=p0+(p1-p0)*x/Image.Width-(p1-p0);
            y0:=q0+(q1-q0)*y/Image.Height-(q1-q0);
            x1:=p0+(p1-p0)*x/Image.Width+(p1-p0);
            y1:=q0+(q1-q0)*y/Image.Height+(q1-q0);
        end;
    edit1.Text:=FloatToStr(x0); edit2.Text:=FloatToStr(x1);
    edit3.Text:=FloatToStr(y0); edit4.Text:=FloatToStr(y1);
    btnDraw.Click
end;
```

7. С помощью сочетания клавиш **Ctrl+F9** выполните компиляцию программы. При необходимости исправьте ошибки.

8. В меню **Project** выбрали пункт **Options**. В отрывшемся диалоге в закладке **Application** загрузите свою иконку, нажав на кнопку **Load Icon**. Нажмите **OK**. В меню **Project** выбрали пункт **Компилировать**. Зайдите в свою папку, найдите откомпилированный exe-файл. Обратите внимание, что файл имеет вашу иконку!! Запустите.

9. **Поэкспериментируйте с разными цветами.** В процедуре, отрабатывающей построение фрактала.

10. **Сохраните несколько фракталов.**

11. **Перепишите (или отсканируйте и вставьте) в тетрадь настоящий документ.**  
Проект завершён.

